

Tema 6: Reglas

Ejemplos

Clasificación de gemas

<i>Gema</i>	<i>Dureza</i>	<i>Densidad</i>	<i>Colores</i>
Diamante	10	3.5	amarillo, marrón, verde, azul, blanco, incoloro
Cuarzo	7	2.65	rojo, rosa, verde, azul, violeta, blanco, negro, incoloro
Turmalina	7	3.1	rojo, rosa, amarillo, marrón, verde, azul, blanco, negro, incoloro
Topacio	8	3.5	rojo, rosa, amarillo, marrón, azul, violeta, blanco, incoloro
Circonio	7	4.7	amarillo, marrón, verde, violeta, blanco, incoloro
Jade	7	3.3	rojo, rosa, amarillo, marrón, verde, azul, violeta, blanco, negro, incoloro
Ópalo	6	2.5	rojo, rosa, amarillo, marrón, blanco, negro, incoloro
Turquesa	5.5	2.5	azul

Clasificación de gemas

Solución 1

```
(defrule diamante
  (dureza 10)
  (densidad 3.5)
  (color amarillo | marron | verde | azul |
   blanco | incoloro)
=>
  (printout t "Diamante" crlf))

(defrule cuarzo
  (dureza 7)
  (densidad 2.65)
  (color rojo | rosa | verde | azul | violeta |
   blanco | negro | incoloro)
=>
  (printout t "Cuarzo" crlf))
```

Clasificación de gemas

Solución 1

```
(defrule turmalina
  (dureza 7)
  (densidad 3.1)
  (color rojo | rosa | amarillo | marron | verde
   | azul | blanco | negro | incoloro)
=>
  (printout t "Turmalina" crlf))
```

```
(defrule topacio
  (dureza 8)
  (densidad 3.5)
  (color rojo | rosa | amarillo | marron | azul |
   violeta | blanco | incoloro)
=>
  (printout t "Topacio" crlf))
```

Clasificación de gemas

Solución 1

```
(defrule circonio
  (dureza 7)
  (densidad 4.7)
  (color amarillo | marron | verde | violeta |
   blanco | incoloro)
=>
  (printout t "Circonio" crlf))
```

```
(defrule jade
  (dureza 7)
  (densidad 3.3)
  (color rojo | rosa | amarillo | marron | verde
   | azul | violeta | blanco | negro | incoloro)
=>
  (printout t "Jadeita" crlf))
```

Clasificación de gemas

Solución 1

```
(defrule opalo
  (dureza 6)
  (densidad 2.5)
  (color rojo | rosa | amarillo | marron | blanco
   | negro | incoloro)
=>
  (printout t "Ópalo" crlf))
```

```
(defrule turquesa
  (dureza 5.5)
  (densidad 2.5)
  (color azul)
=>
  (printout t "Turquesa" crlf))
```

Clasificación de gemas

Solución 2

```
(deftemplate gema  
  (slot nombre)  
  (slot dureza)  
  (slot densidad)  
  (multislot colores))
```

Clasificación de gemas

Solución 2

(deffacts gemas

(gema (nombre diamante) (dureza 10) (densidad 3.5) (colores amarillo marron verde azul blanco incoloro))

(gema (nombre cuarzo) (dureza 7) (densidad 2.65) (colores rojo rosa verde azul violeta blanco negro incoloro))

(gema (nombre turmalina) (dureza 7) (densidad 3.1) (colores rojo rosa amarillo marron verde azul blanco negro incoloro))

.
.
.)

Clasificación de gemas

Solución 2

```
(defrule identificar-gema
  (dureza ?du)
  (densidad ?de)
  (color ?c)
  (gema (nombre ?n) (dureza ?du) (densidad ?de)
    (colores $? ?c $?))
=>
  (printout t ?n crlf))
```